Regge-Wheeler-ova rovnice
Přepis do prvního řádu v čase

In[ ]:=
```
∂tt ψ == ∂xx ψ - V ψ // HoldForm
∂t MatrixForm[{ψ, χ}] == MatrixForm[{χ, ∂xx ψ - V ψ}] // HoldForm
```

Out[ ]= $\partial_{tt} \psi == \partial_{xx} \psi - V\ \psi$

Out[ ]= $\partial_t \begin{pmatrix} \psi \\ \chi \end{pmatrix} == \begin{pmatrix} \chi \\ \partial_{xx} \psi - V\ \psi \end{pmatrix}$

želví souřadnice

In[1]:=
```
rStar == r + 2 M Log[r / (2 M) - 1]
Solve[%, r][[1]]
rInv[rStar_] = Evaluate@(r /. % /. M → 1)
rInv[-10.]
```

Out[1]= $\text{rStar} == r + 2\ M\ \text{Log}\left[-1 + \frac{r}{2\ M}\right]$

Solve: Inverse functions are being used by Solve, so some solutions may not be found; use Reduce for complete solution information.

Out[2]= $\left\{r \to 2\left(M + M\ \text{ProductLog}\left[e^{-1+\frac{\text{rStar}}{2\ M}}\right]\right)\right\}$

Out[3]= $2\left(1 + \text{ProductLog}\left[e^{-1+\frac{\text{rStar}}{2}}\right]\right)$

Out[4]= 2.00495

Konkrétní potenciál pro numerické řešení

In[5]:=
```
params = {M → 1, s → 2, l → 2}
V_RW[r_] = (1 - 2 M / r) (l (l + 1) / r^2 + 2 M / r^3 (1 - s^2)) /. params
```

Out[5]= {M → 1, s → 2, l → 2}

Out[6]= $\left(-\frac{6}{r^3} + \frac{6}{r^2}\right)\left(1 - \frac{2}{r}\right)$

In[7]:=
```
Row[{ Plot[ V_RW[r], {r, 2, 10}, ImageSize → Medium, PlotTheme → "Scientific"],
  Plot[ V_RW[rInv[x]], {x, -8, 10}, ImageSize → Medium, PlotTheme → "Scientific"]}]
```

Out[7]=





Pravá strana rovnice

$$\partial_t \begin{pmatrix} \psi \\ \chi \end{pmatrix} == \begin{pmatrix} \chi \\ \partial_{xx} \psi - V \psi \end{pmatrix}$$

Numerický integrátor pro MOL

```
In[8]:= rhs := Compile[{{x, _Real, 1}, {V, _Real, 1}, {u, _Real, 2}},
   Module[{ i, j, f, dx, n = Length[u], u0, iψ = 1, iχ = 2},
    dx = x[[2]] - x[[1]];
    u0 = {0., 0.};
    f = u ;
    f[[1]] = {u[[1, iχ]], 0};
    f[[-1]] = {u[[n, iχ]], 0};
    For[i = 2, i ≤ n - 1, i++,
     f[[i, 1]] = u[[i, 2]];
     f[[i, 2]] = (u[[i - 1, 1]] - 2 * u[[i, 1]] + u[[i + 1, 1]]) / (dx ^ 2) - V[[i]] × u[[i, 1]];
    ];
    f
   ]
   (*,CompilationOptions→{"InlineExternalDefinitions"→True}*)];

In[ ]:= (*<<"CompiledFunctionTools`"
 CompilePrint[rhs]*)

In[9]:= (* Heun's RK3, no explicit time in rhs *)
RK3Step[u_, dt_] := Block[{u1, u2, u3, dt3},
  dt3 = dt / 3.;
  u1 = u + dt3 rhs[xDomain, Vx, u];
  u2 = u + 2 dt3 rhs[xDomain, Vx, u1];
  u3 = u1 + dt rhs[xDomain, Vx, u2];
  0.75 u3 + 0.25 u
 ]
```

Počáteční prostorový tvar vlny
Parametry gridu
Předpočítané hodnoty potenciálu

```
In[10]:= ψ0[x_Real] := If[x > 3 && x < 6, {Sin[Pi x / 3] ^ 2, 0}, {0, 0}]

nDomain = 5000;
xBdry = 100.;
xDomain = Range[-xBdry, xBdry, 2 xBdry / nDomain];
Vx = V_RW[rInv[#]] & /@ xDomain;
uInit = ψ0 /@ xDomain;
iObserver = 11 / 20 * nDomain
rObserver == rInv[xDomain[[iObserver]]]

Out[16]= 2750

Out[17]= rObserver == 7.82275
```

Inicializace

Hlavní smyčka řešení pomocí MOL

In[18]:=
```
tFin = 94;
dtPrint = 0.5;

cfl = 0.5;
t0 = 0;
dx0 = xDomain[[2]] - xDomain[[1]];
dt0 = cfl * dx0
u0 = uInit;

tPrint = t0;
results = {}; observed = {};
(*u0
  u0 = RK3Step[ xDomain, u0,dt0]
*)
Do[
 If[ t0 ≥ tPrint, AppendTo[ results, {t0, u0}]; tPrint += dtPrint];
 AppendTo[ observed, {t0, u0[[iObserver]]}];
 u0 = RK3Step[ u0, dt0];
 t0 += dt0;
 , {iStep, 1, Round[tFin / dt0]}]
```

Out[23]= 0.02

Jaké ψ vidí pozorovatel na fixním r

In[28]:=
```
ω022 = 0.37367 - 0.08896 ⅈ;
ω122 = 0.34671 - 0.27391 ⅈ;
(*podle https://www.ncbi.nlm.nih.gov/pmc/articles/PMC5253841/  *)
Show[
 ListLinePlot[Transpose@{observed[[ ;; , 1]], observed[[ ;; , 2, 1]]},
  PlotTheme → "Scientific", PlotRange → {-0.21, 0.21}, ImageSize → Large],
 Plot[ Re[ 0.41 Exp[-ⅈ ω022 (t - 0.9)]], {t, 15, 80}, PlotRange → Full]
]
```

Out[30]=



Problém: Ukažte, že až na amplitudu a fázový posun je pro různá počáteční data (např. různý tvar, poloha nebo počet vlnek) tzv. “ringdown” stejný (jde o dominantní QNM)

Animace časové závislosti prostorového průběhu ψ

In[31]:=
```
Table[
  ListLinePlot[Transpose@{xDomain, results[[k, 2, ;; , 1]]} ,
   PlotRange → {-1, 1}, PlotTheme → "Scientific",
   PlotLabel → "t=" <> ToString@results[[k, 1]], ImageSize → 666],
  {k, 1, Length@results}
 ];
ListAnimate@%
```

Out[32]=



In[ ]:=